The MEANS/SUMMARY Procedure: 
Getting Started

Arthur L. Carpenter
California Occidental Consultants

ABSTRACT
The MEANS/SUMMARY procedure is a workhorse for most data analysts. It is used to create tables of summary statistics as well as complex summary data sets. The user has a great many options which can be used to customize what the procedure is to produce. Unfortunately most analysts rely on only a few of the simpler basic ways of setting up the PROC step, never realizing that a number of less commonly used options and statements exist that can greatly simplify the procedure code, the analysis steps, and the resulting output.

This tutorial begins with the basic statements of the MEANS/SUMMARY procedure and follows up with introductions to a number of important and useful options and statements that can provide the analyst with much needed tools. With this practical knowledge, you can greatly enhance the usability of the procedure and then you too will be doing more with MEANS/SUMMARY.

KEY WORDS
OUTPUT, MEANS, SUMMARY, AUTONAME, _TYPE_, WAYS, LEVELS, MAXID, GROUPID, preloaded formats

INTRODUCTION
PROC MEANS is one of SAS’s original procedures, and it’s initial mandate was to create printed tables of summary statistics. Later PROC SUMMARY was introduced to create summary data sets. Although these two procedures grew up on the opposite side of the tracks, over time both has evolved so that under the current version of SAS they actually both use the same software behind the scenes.

These two procedures completely share capabilities. In fact neither can do anything that the other cannot do. Only some of the defaults are different (as they reflect the procedures’ original roots).

For the analyst faced with creating statistical summaries, the MEANS/SUMMARY procedure is indispensable. While it is fairly simple to generate a straightforward statistical summary, these procedures allow a complex list of options and statements that give the analyst a great deal of control.

Because of the similarity of these two procedures, examples will tend to show one or the other but not both. When I use MEANS or SUMMARY, I tend to select the procedure based on it primary objective of the step (SUMMARY for a summary data set and MEANS for a printed table). Even that ‘rule’, however is rather lax as MEANS has the further advantage of only having 5 letters in the procedure name.
**BASIC STATEMENTS**
The MEANS/SUMMARY procedure is so powerful that just a few simple statements and options can produce fairly complex and useful summary tables.

**Differences Between MEANS and SUMMARY**
Originally MEANS was used to generate printed tables and SUMMARY a summary data set. While both procedures can now create either type of output, the defaults for both tend to reflect the original roots of the procedure.

One of the primary differences in defaults is seen by looking at the way each procedure creates printed tables. Printed tables are routed through the Output Delivery System to a destination such as LISTING or HTML. By default MEANS always creates a table to be printed. If you do not want a printed table you must explicitly turn it off (NOPRINT option). On the other hand, the SUMMARY procedure never creates a printed table unless it is specifically requested (PRINT option).

There are a few other differences between MEANS and SUMMARY. In each case the difference reflects default behaviors, and these will be pointed out in the appropriate sections of this paper.

**Creating a Basic Summary Table**
Very little needs to be done to create a simple summary table. The DATA= option in the PROC statement identifies the data set to be summarized and the VAR statement lists one or more numeric variables to be analyzed.

```sas
proc means data=sashelp.class;
var weight;
run;
```

We can see that the mean weight of the 19 students in the CLASS data set is something over 100 pounds. Because we left the selection of the statistics to the defaults, the table contains N, mean, standard deviation, minimum and the maximum.

**Selecting Statistics**
Generally we want more control over which statistics are to be selected. When you want to specifically select statistics, they are listed as options on the PROC statement.

```sas
title1 'The First Two Statistical Moments';
proc means data=sashelp.class n mean var std stderr;
var weight;
run;
```

We can see that the mean weight of the 19 students in the CLASS data set is something over 100 pounds. Because we left the selection of the statistics to the defaults, the table contains N, mean, standard deviation, minimum and the maximum.
The list of available statistics is fairly comprehensive. A subset of which includes:

- **n**  number of observations used to calculate the statistics
- **nmiss**  number of observations with missing values
- **min**  minimum value taken on by the data
- **max**  maximum value taken on by the data
- **range**  difference between the min and the max
- **sum**  total of the data
- **mean**  arithmetic mean
- **std**  standard deviation
- **stderr**  standard error
- **var**  variance
- **skewness**  symmetry of the data's distribution
- **kurtosis**  peakedness of the data's distribution

A number of statistics having to do with percentiles and quantiles are also available, including:

- **median**  50th percentile
- **p50**  50th percentile (or second quartile)
- **p25 | q1**  25th percentile (or first quartile)
- **p75 | q3**  75th percentile (or third quartile)
- **p1 p5 p10**  other percentiles
- **p90 p95 p99**  other percentiles

Starting in SAS9.2 the MODE statistic is also available.

Statistics listed on the PROC statement are only applied to the printed table and have **NOTHING** to do with and summary data sets that are also created.

### Creating a Summary Data Set

Both procedures can also be used to create a summary data set through the use of the OUTPUT statement. Without using ODS, a summary data set will not be created unless the OUTPUT statement is present. This is true for both the MEANS and SUMMARY procedures.

```sas
title1 'A Simple Summary Data Set';
proc means data=sashelp.class
   noprint;
   var weight;
   output out=summrydat;
run;
```

The NOPRINT option is used with MEANS, because a printed table is not wanted. A PROC PRINT of the summary data set (WORK.SUMMRYDAT) shows the following:

<table>
<thead>
<tr>
<th>Obs</th>
<th><em>TYPE</em></th>
<th><em>FREQ</em></th>
<th>STAT_</th>
<th>Weight</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>0</td>
<td>19</td>
<td>N</td>
<td>19.000</td>
</tr>
<tr>
<td>2</td>
<td>0</td>
<td>19</td>
<td>MIN</td>
<td>50.500</td>
</tr>
<tr>
<td>3</td>
<td>0</td>
<td>19</td>
<td>MAX</td>
<td>150.000</td>
</tr>
<tr>
<td>4</td>
<td>0</td>
<td>19</td>
<td>MEAN</td>
<td>100.026</td>
</tr>
<tr>
<td>5</td>
<td>0</td>
<td>19</td>
<td>STD</td>
<td>22.774</td>
</tr>
</tbody>
</table>

Again since statistics were not specified the same default list of statistics as was used in the MEANS’s printed table appears here.

### Selecting the Statistics and Naming the Variables in the Summary Data Set

Usually when you create a summary data set, you will want to specifically select the statistics. These are specified on the OUTPUT statement. Remember statistics listed on the PROC statement only apply to printed tables and have nothing to do with the statistics that you want in the summary data set.

The techniques shown below can be combined - experiment.

#### Selecting Statistics

Statistics are selected by using their names as options in the OUTPUT statement. The name of each statistic is followed by an equal sign. The following OUTPUT statement requests that the mean weight be calculated and saved in the data set SUMMRYDAT.
title1 'Selected Statistics';
proc summary data=sashelp.class;
var weight;
output out=summrydat mean=;
run;

The mean weight will be stored in a variable named WEIGHT. This technique allows you to only pick a single statistic, and as such it is limited, however when combined with the techniques shown below, it can be very flexible.

**Explicate Naming**

By following the equal sign with a name, you can provide names for the new variables. This allows you to name more than one statistic on the OUTPUT statement.

```
title1 'Selecting Multiple Statistics';
proc summary data=sashelp.class;
var weight;
output out=summrydat n=number mean=average std=std_deviation;
run;
```

You can also name multiple analysis variables. Here both HEIGHT and WEIGHT are specified.

```
title1 'Multiple Analysis Variables';
proc summary data=sashelp.class;
var height weight;
output out =summrydat
    n    = ht_n    wt_n
    mean = mean_ht mean_wt
    std  = sd_ht   sd_wt;
run;
```

Be sure to be careful here as the order of the variables in the VAR statement determines which variable is for height and which is for weight. You should also be smart about naming conventions. In the previous example the statistics for N are not consistently named relative to those for the MEAN and STD.

This technique does not allow you to ‘skip’ statistics. If you did not want the mean for HEIGHT, but only the mean for WEIGHT, this would not be possible, because HEIGHT is first on the VAR statement. To get around this you can use the techniques on naming the statistics shown in the next section.
Selected Naming
When there is more than one variable in the VAR statement, but you do not want every statistic calculated for every analysis variable, you can selectively associate statistics with analysis variables.

```
title1 'Selective Associations';
proc summary data=sashelp.class;
var height weight;
output out =summrydat
   n   =ht_n  wt_n
   mean(weight)= wt_mean
   std(height) = ht_std;
run;
```

Alternate forms of the statistic selections (in this case for the MEAN) could have included the following:

```
mean(weight height)=wt_mean ht_mean
mean(weight)=wt_mean
mean(height)=ht_mean
```

Automatic Naming of Summary Variables
When you do not NEED to control the naming of the new summary variables, the AUTONAME and AUTOLABEL options can be used on the OUTPUT statement.

The AUTONAME option allows you to select statistics without picking a name for the resulting variable in the OUTPUT table. This eliminates naming conflicts. The AUTOLABEL option creates a label for variables added to the OUT= data set.

```
title1 'Using AUTONAME';
proc summary data=sashelp.class;
var height weight;
output out =summrydat
   n   =
   mean=
   std = / autoname;
run;
```

```
Using AUTONAME

<table>
<thead>
<tr>
<th>Obs</th>
<th><em>TYPE</em></th>
<th><em>FREQ</em></th>
<th>Height_N</th>
<th>Weight_N</th>
<th>Height Mean</th>
<th>Weight Mean</th>
<th>Height StdDev</th>
<th>Weight StdDev</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>0</td>
<td>19</td>
<td>19</td>
<td>19</td>
<td>62.3368</td>
<td>100.026</td>
<td>5.12708</td>
<td>22.7739</td>
</tr>
</tbody>
</table>
```

Notice that the names are in the form of `variable_statistic`. This is a nicely consistent, dependable, and usable naming convention.
Using the CLASS Statement
The CLASS statement can be used to create subgroups. Unlike the BY statement the data do not have to be sorted prior to its use. Like in most other procedures that utilize the CLASS statement, there can be one or more classification variables.

In a Printed Table
When the resulting table is to be printed, CLASS creates one summary for each combination of classification variables.

```
title1 'CLASS and a Printed Table';
proc means data=sashelp.class(where=(age in(12,13,14)))
  n mean std;
class age sex;
var height;
run;
```

In a Summary Data Set
When creating a summary data set, one can get not only the classification variable interaction statistics, but the main factor statistics as well. This can be very helpful to the statistician.

```
title1 'CLASS and a Summary Data Set';
proc summary data=sashelp.class(where=(age in(12,13,14)));
class age sex;
var height;
output out=clsummry n=ht_n mean=ht_mean std=ht_sd;
run
```

A PROC PRINT of the data set CLSUMMRY shows:

```
CLASS and a Summary Data Set

<table>
<thead>
<tr>
<th>Obs</th>
<th>Age</th>
<th>Sex</th>
<th><em>TYPE</em></th>
<th><em>FREQ</em></th>
<th>ht_n</th>
<th>ht_mean</th>
<th>ht_sd</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>.</td>
<td>.</td>
<td>0</td>
<td>12</td>
<td>12</td>
<td>61.7583</td>
<td>3.97868</td>
</tr>
<tr>
<td>2</td>
<td>.</td>
<td>F</td>
<td>1</td>
<td>6</td>
<td>6</td>
<td>60.8333</td>
<td>3.90470</td>
</tr>
<tr>
<td>3</td>
<td>.</td>
<td>M</td>
<td>1</td>
<td>6</td>
<td>6</td>
<td>62.6833</td>
<td>4.18637</td>
</tr>
<tr>
<td>4</td>
<td>12</td>
<td>2</td>
<td>5</td>
<td>5</td>
<td>5</td>
<td>59.4400</td>
<td>3.29742</td>
</tr>
<tr>
<td>5</td>
<td>13</td>
<td>2</td>
<td>3</td>
<td>3</td>
<td>3</td>
<td>61.4333</td>
<td>4.49592</td>
</tr>
<tr>
<td>6</td>
<td>14</td>
<td>2</td>
<td>4</td>
<td>4</td>
<td>4</td>
<td>64.9000</td>
<td>2.80119</td>
</tr>
<tr>
<td>7</td>
<td>12</td>
<td>F</td>
<td>3</td>
<td>2</td>
<td>2</td>
<td>63.5500</td>
<td>1.06066</td>
</tr>
<tr>
<td>8</td>
<td>12</td>
<td>M</td>
<td>3</td>
<td>3</td>
<td>3</td>
<td>60.3667</td>
<td>3.88909</td>
</tr>
</tbody>
</table>
```

Two additional variables have been added to the summary data set; _TYPE_ (which is described below in more detail), and _FREQ_ (which counts observations). Although not apparent in this example, _FREQ_ counts all observations, while the N
Understanding _TYPE_

The _TYPE_ variable in the output data set helps us track the level of summarization, and can be used to distinguish the sets of statistics. Notice in the previous example that _TYPE_ changes for each level of summarization.

<table>
<thead>
<tr>
<th><em>TYPE</em></th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>0</td>
<td>Summarize across all classification variables</td>
</tr>
<tr>
<td>1</td>
<td>Summarize as if the right most classification variable (SEX) was the only one</td>
</tr>
<tr>
<td>2</td>
<td>Summarize as if the next to the right most classification variable (AGE) was the only one</td>
</tr>
<tr>
<td>3</td>
<td>Interaction of the two classification variables.</td>
</tr>
</tbody>
</table>

In the following example there are three CLASS variables and _TYPE_ ranges from 0 to 7.

title1 'Understanding _TYPE_';
proc summary data=advrpt.demog(where=(race in('1','4')
                                   & 12 le edu le 15
                                   & symp in('01','02','03')));
class race edu symp;
var ht;
output out=stats mean= meanHT;
run;

<table>
<thead>
<tr>
<th>Obs</th>
<th>RACE</th>
<th>EDU</th>
<th>SYMP</th>
<th><em>TYPE</em></th>
<th><em>FREQ</em></th>
<th>HT</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>.</td>
<td>.</td>
<td>.</td>
<td>0</td>
<td>8</td>
<td>66.25</td>
</tr>
<tr>
<td>2</td>
<td>.</td>
<td>01</td>
<td>.</td>
<td>1</td>
<td>2</td>
<td>64.00</td>
</tr>
<tr>
<td>3</td>
<td>.</td>
<td>02</td>
<td>.</td>
<td>1</td>
<td>4</td>
<td>66.50</td>
</tr>
<tr>
<td>4</td>
<td>.</td>
<td>03</td>
<td>.</td>
<td>1</td>
<td>2</td>
<td>68.00</td>
</tr>
<tr>
<td>5</td>
<td>12</td>
<td>2</td>
<td>.</td>
<td>4</td>
<td>2</td>
<td>64.00</td>
</tr>
<tr>
<td>6</td>
<td>14</td>
<td>2</td>
<td>.</td>
<td>4</td>
<td>2</td>
<td>67.50</td>
</tr>
<tr>
<td>7</td>
<td>15</td>
<td>2</td>
<td>.</td>
<td>4</td>
<td>2</td>
<td>66.00</td>
</tr>
<tr>
<td>8</td>
<td>12</td>
<td>02</td>
<td>3</td>
<td>2</td>
<td>67.00</td>
<td></td>
</tr>
<tr>
<td>9</td>
<td>12</td>
<td>03</td>
<td>3</td>
<td>2</td>
<td>68.00</td>
<td></td>
</tr>
<tr>
<td>10</td>
<td>14</td>
<td>01</td>
<td>3</td>
<td>2</td>
<td>64.00</td>
<td></td>
</tr>
<tr>
<td>11</td>
<td>15</td>
<td>02</td>
<td>3</td>
<td>2</td>
<td>66.00</td>
<td></td>
</tr>
<tr>
<td>12</td>
<td>1</td>
<td>.</td>
<td>4</td>
<td>6</td>
<td>67.00</td>
<td></td>
</tr>
<tr>
<td>13</td>
<td>4</td>
<td>.</td>
<td>4</td>
<td>2</td>
<td>64.00</td>
<td></td>
</tr>
<tr>
<td>14</td>
<td>1</td>
<td>.</td>
<td>02</td>
<td>5</td>
<td>4</td>
<td>66.50</td>
</tr>
<tr>
<td>15</td>
<td>1</td>
<td>.</td>
<td>03</td>
<td>5</td>
<td>2</td>
<td>68.00</td>
</tr>
<tr>
<td>16</td>
<td>4</td>
<td>.</td>
<td>01</td>
<td>5</td>
<td>2</td>
<td>64.00</td>
</tr>
<tr>
<td>17</td>
<td>1</td>
<td>12</td>
<td>6</td>
<td>4</td>
<td>67.50</td>
<td></td>
</tr>
<tr>
<td>18</td>
<td>1</td>
<td>15</td>
<td>6</td>
<td>2</td>
<td>66.00</td>
<td></td>
</tr>
<tr>
<td>19</td>
<td>4</td>
<td>14</td>
<td>6</td>
<td>2</td>
<td>64.00</td>
<td></td>
</tr>
<tr>
<td>20</td>
<td>1</td>
<td>12</td>
<td>02</td>
<td>7</td>
<td>2</td>
<td>67.00</td>
</tr>
<tr>
<td>21</td>
<td>1</td>
<td>12</td>
<td>03</td>
<td>7</td>
<td>2</td>
<td>68.00</td>
</tr>
<tr>
<td>22</td>
<td>1</td>
<td>15</td>
<td>02</td>
<td>7</td>
<td>2</td>
<td>66.00</td>
</tr>
<tr>
<td>23</td>
<td>4</td>
<td>14</td>
<td>01</td>
<td>7</td>
<td>2</td>
<td>64.00</td>
</tr>
</tbody>
</table>

statistic only counts observations with non-missing values.

If you only want the statistics for the highest order interaction, you can use the NWAY option on the PROC statement.

```plaintext
proc summary data=sashelp.class(where=(age in(12,13,14)))
nway;
```
When calculating the value of _TYPE_, assign a zero (0) when summarizing over a CLASS variable and assign a one (1) when summarizing for the CLASS variable. In the table below the zeros and ones associated with the class variables form a binary value. This binary value can be converted to decimal to obtain _TYPE_.

<table>
<thead>
<tr>
<th>Observations</th>
<th>RACE</th>
<th>EDU</th>
<th>SYMP</th>
<th>Binary Value</th>
<th><em>TYPE</em></th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>0</td>
<td>0</td>
<td>0</td>
<td>000</td>
<td>0</td>
</tr>
<tr>
<td>2 - 4</td>
<td>0</td>
<td>0</td>
<td>1</td>
<td>001</td>
<td>1</td>
</tr>
<tr>
<td>5 - 7</td>
<td>0</td>
<td>1</td>
<td>0</td>
<td>010</td>
<td>2</td>
</tr>
<tr>
<td>8 - 11</td>
<td>0</td>
<td>1</td>
<td>1</td>
<td>011</td>
<td>3</td>
</tr>
<tr>
<td>12 - 13</td>
<td>1</td>
<td>0</td>
<td>0</td>
<td>100</td>
<td>4</td>
</tr>
<tr>
<td>14 - 16</td>
<td>1</td>
<td>0</td>
<td>1</td>
<td>101</td>
<td>5</td>
</tr>
<tr>
<td>17 - 19</td>
<td>1</td>
<td>1</td>
<td>0</td>
<td>110</td>
<td>6</td>
</tr>
<tr>
<td>20 - 23</td>
<td>1</td>
<td>1</td>
<td>1</td>
<td>111</td>
<td>7</td>
</tr>
</tbody>
</table>

2^4 = 16, 2^3 = 8, 2^2 = 4, 2^1 = 2

A binary value of 110 = 1*2^2 + 1*2^1 + 0*2^0 = 1*4 + 1*2 + 0*1 = 6 = _TYPE_

Some SAS programmers find converting binary values to decimal values a bit tedious. Fortunately the developers at SAS Institute have provided us with alternatives.

**Using CHARTYPE**

The CHARTYPE option causes _TYPE_ to be displayed as a character variable in binary form rather than as a decimal value.

```sas
ods html;
proc means data=advrvpt.demog(where=(race in('1','4')
& 12 le edu le 15
& symp in('01','02','03'))
chartype; class race edu symp; var ht; output out=stats mean= meanHT; run;
ods html close;
```

```sas
title1 'Understanding _TYPE_ Using CHARTYPE';
proc summary data=advrvpt.demog(where=(race in('1','4')
& 12 le edu le 15
& symp in('01','02','03'))
chartype; class race edu symp; var ht; output out=stats mean= meanHT; run;
```
Understanding _TYPE_ Using CHARTYPE

<table>
<thead>
<tr>
<th>Obs</th>
<th>RACE</th>
<th>EDU</th>
<th>SYMP</th>
<th><em>TYPE</em></th>
<th><em>FREQ</em></th>
<th>HT</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>.</td>
<td>.</td>
<td>000</td>
<td>8</td>
<td>66.25</td>
<td></td>
</tr>
<tr>
<td>2</td>
<td>.</td>
<td>01</td>
<td>001</td>
<td>2</td>
<td>64.00</td>
<td></td>
</tr>
<tr>
<td>3</td>
<td>.</td>
<td>02</td>
<td>001</td>
<td>4</td>
<td>66.50</td>
<td></td>
</tr>
<tr>
<td>4</td>
<td>.</td>
<td>03</td>
<td>001</td>
<td>2</td>
<td>68.00</td>
<td></td>
</tr>
<tr>
<td>5</td>
<td>12</td>
<td>010</td>
<td>4</td>
<td>67.50</td>
<td></td>
<td></td>
</tr>
<tr>
<td>6</td>
<td>14</td>
<td>010</td>
<td>2</td>
<td>64.00</td>
<td></td>
<td></td>
</tr>
<tr>
<td>7</td>
<td>15</td>
<td>010</td>
<td>2</td>
<td>66.00</td>
<td></td>
<td></td>
</tr>
<tr>
<td>8</td>
<td>12</td>
<td>02</td>
<td>011</td>
<td>2</td>
<td>67.00</td>
<td></td>
</tr>
<tr>
<td>9</td>
<td>12</td>
<td>03</td>
<td>011</td>
<td>2</td>
<td>68.00</td>
<td></td>
</tr>
</tbody>
</table>

. . . . portions of the table not shown . . . .

CREATING SUMMARY DATA SUBSETS
Once you have started to create summary data sets with MEANS/SUMMARY, you will soon discover how very useful they can be. Of course you will often find that you do not need all the information contained in the summary data set and that you need to create a data subset. As with most things in SAS there are multiple ways to do this. We have already seen the use of the NWAY option to subset for only the highest order interaction. This is fine but not very flexible. Let’s look at some techniques that are a bit more useful.

Select Rows Using _TYPE_
Once you understand and can predict the value of _TYPE_, it can be used to provide subsetting information in a followup DATA step. Suppose that in the previous example we would like to have only those rows for which EDU is a factor. Our DATA step might be written something like:

```sas
proc summary data=advrpt.demog(where=(race in('1','4')
& 12 le edu le 15
& symp in('01','02','03')));
class race edu symp;
var ht;
output out=stats mean= meanHT;
run;

data edufactor;
set stats;
where _type_ in(2,3,6,7);
run;
```

Better yet why not avoid the DATA step and build the subset directly when the data are created.

```sas
proc summary data=advrpt.demog(where=(race in('1','4')
& 12 le edu le 15
& symp in('01','02','03')));
class race edu symp;
var ht;
output out=edufactor(where=(_type_ in(2,3,6,7)))
    mean= meanHT;
run;
```
SUMMARY
The MEANS /SUMMARY procedure produces a wide variety of summary reports and summary data tables. It is very flexible and, while it can be quite complex, a few basic statements allow the user to create useful summaries.

As you develop a deeper knowledge of the MEANS/SUMMARY procedure, you will find that the generation of highly sophisticated summarizations is possible from within a single step.

ABOUT THE AUTHOR
Art Carpenter’s publications list includes four books, and numerous papers and posters presented at SUGI, SAS Global Forum, and other user group conferences. Art has been using SAS® since 1976 and has served in various leadership positions in local, regional, national, and international user groups. He is a SAS Certified Professional™ and through California Occidental Consultants he teaches SAS courses and provides contract SAS programming support nationwide.

AUTHOR CONTACT
Arthur L. Carpenter
California Occidental Consultants
10606 Ketch Circle
Anchorage, AK 99515
(907) 865-9167
art@caloxy.com
www.caloxy.com

TRADEMARK INFORMATION
SAS, SAS Certified Professional, SAS Certified Advanced Programmer, and all other SAS Institute Inc. product or service names are registered trademarks of SAS Institute, Inc. in the USA and other countries.
® indicates USA registration.